
	Fixing errors found in 2014wk24 release for TC 15.11

Change 1 – f_IMS_MOCallHoldResume_AnnexC8
	Function name
	 f_IMS_MOCallHoldResume_AnnexC8

	Reason for change
	In function fl_IMS_BuildSDP_AnnexC8_Step1_Attributes(), see Change 3, we need the information whether we are in the scenario of Call Hold or Call Resume. As the two scenarios are entered from the main function f_TC_15_11_IMS() via function f_IMS_MOCallHoldResume_AnnexC8(), we need to forward above information from there until  fl_IMS_BuildSDP_AnnexC8_Step1_Attributes(). 

	Summary of change
	Forward the information from the argument list of f_IMS_MOCallHoldResume_AnnexC8 to by adding p_HoldResume of type IMS_CallHoldResume_Type to argument list of embedded function f_IMS_BuildSDP_AnnexC8_Step1.

	TTCN module
	 Common/IMS/IMS_Procedures_CallControl

	MCC160 Comment
	


Before change

    function f_IMS_MOCallHoldResume_AnnexC8(IMS_CallHoldResume_Type p_HoldResume,

                                          INVITE_Request p_PrevInviteRequest,

                                          MessageHeader p_MessageHeader_Prev183Response,

                                          SDP_Message p_SDP_OriginalOffer,

                                          template (omit) integer p_PrevSessionVersion := omit) runs on IMS_PTC return SDP_Message

  {

…

    v_SDP_HoldResumeOffer := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_MessageBody, f_IMS_BuildSDP_AnnexC8_Step1(p_SDP_OriginalOffer));  /* @sic R5s140324 - additional change 2 sic@ */

…
  }
After change

     function f_IMS_MOCallHoldResume_AnnexC8(IMS_CallHoldResume_Type p_HoldResume,

                                          INVITE_Request p_PrevInviteRequest,

                                          MessageHeader p_MessageHeader_Prev183Response,

                                          SDP_Message p_SDP_OriginalOffer,

                                          template (omit) integer p_PrevSessionVersion := omit) runs on IMS_PTC return SDP_Message

  {

…
    v_SDP_HoldResumeOffer := f_IMS_MessageBody_CheckSdpMessageAndGeolocation(v_MessageBody, f_IMS_BuildSDP_AnnexC8_Step1(p_SDP_OriginalOffer, p_HoldResume));  /* @sic R5s140324 - additional change 2 sic@ */
…
  }
Change 2 – f_IMS_BuildSDP_AnnexC8_Step1
	Function name
	 f_IMS_BuildSDP_AnnexC8_Step1

	Reason for change
	 See Change 1

	Summary of change
	Continue forwarding the information from f_IMS_BuildSDP_AnnexC8_Step1 to fl_IMS_BuildSDP_AnnexC8_Step1_Attributes

	TTCN module
	 Common/IMS/IMS_SDP_Messages

	MCC160 Comment
	


Before change

    function f_IMS_BuildSDP_AnnexC8_Step1(SDP_Message p_SDP_Message_PrevOffer) runs on IMS_PTC return template (present) SDP_Message

  {

    var template (present) SDP_Message v_SDP_Message := p_SDP_Message_PrevOffer;

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var SDP_media_desc v_MediaDesc;

    var SDP_media_field v_MediaField;

    var integer i;

    v_SDP_Message.origin.session_version := ?;

    v_SDP_Message.bandwidth := fl_IMS_BuildSDP_AnnexC8_Step1_Bandwidth(p_SDP_Message_PrevOffer.bandwidth);

    v_SDP_Message.attributes := fl_IMS_BuildSDP_AnnexC8_Step1_Attributes(p_SDP_Message_PrevOffer.attributes);

    for (i:=0; i < lengthof(p_SDP_Message_PrevOffer.media_list); i:=i+1) {

      v_MediaDesc := p_SDP_Message_PrevOffer.media_list[i];

      v_MediaField := v_MediaDesc.media_field;

      v_SDP_Message.media_list[i] := cr_SDP_Media_Desc(cr_SDP_Media(v_MediaField.media, v_MediaField.ports.port_number, v_MediaField.transport, cr_SDP_FmtList_AtLeastOne(cr_SDP_AnyMediaFormatDescr)),

                                                       cr_SDP_Connection(v_AddrType, ?),

                                                       fl_IMS_BuildSDP_AnnexC8_Step1_Bandwidth(v_MediaDesc.bandwidth),

                                                       fl_IMS_BuildSDP_AnnexC8_Step1_Attributes(v_MediaDesc.attributes));

    }

    return v_SDP_Message;

  }
After change

  function f_IMS_BuildSDP_AnnexC8_Step1(SDP_Message p_SDP_Message_PrevOffer, IMS_CallHoldResume_Type p_HoldResume := HOLD) runs on IMS_PTC return template (present) SDP_Message

  {

    var template (present) SDP_Message v_SDP_Message := p_SDP_Message_PrevOffer;

    var charstring v_AddrType := f_IMS_PTC_NW_Address_GetTypeStr();

    var SDP_media_desc v_MediaDesc;

    var SDP_media_field v_MediaField;

    var integer i;

    v_SDP_Message.origin.session_version := ?;

    v_SDP_Message.bandwidth := fl_IMS_BuildSDP_AnnexC8_Step1_Bandwidth(p_SDP_Message_PrevOffer.bandwidth);

    v_SDP_Message.attributes := fl_IMS_BuildSDP_AnnexC8_Step1_Attributes(p_SDP_Message_PrevOffer.attributes);

    for (i:=0; i < lengthof(p_SDP_Message_PrevOffer.media_list); i:=i+1) {

      v_MediaDesc := p_SDP_Message_PrevOffer.media_list[i];

      v_MediaField := v_MediaDesc.media_field;

      v_SDP_Message.media_list[i] := cr_SDP_Media_Desc(cr_SDP_Media(v_MediaField.media, v_MediaField.ports.port_number, v_MediaField.transport, cr_SDP_FmtList_AtLeastOne(cr_SDP_AnyMediaFormatDescr)),

                                                       cr_SDP_Connection(v_AddrType, ?),

                                                       fl_IMS_BuildSDP_AnnexC8_Step1_Bandwidth(v_MediaDesc.bandwidth),

                                                       fl_IMS_BuildSDP_AnnexC8_Step1_Attributes(v_MediaDesc.attributes, p_HoldResume));  

    }

    return v_SDP_Message;

  }   
Change 3 – fl_IMS_BuildSDP_AnnexC8_Step1_Attributes
	Function name
	 fl_IMS_BuildSDP_AnnexC8_Step1_Attributes

	Reason for change
	This function is used to specify the SDP expected from the UE in steps 1 and 5 of C.8, first for Call Hold, and then for Call Resume. Input to this function is the list of attributes on media level from the SDP last sent by the UE (either in PRACK or in UPDATE of MO call setup).
Following errors were found in the SDP template:

1. The information if this is for Call Hold or Call Resume needs to be input to this function.
2. During MO call setup, the UE might have sent an “a=conf:...” attribute line – but it is not required to do so, see e.g. C.21, step 5. If it does so nonetheless, there is no reason to mandate to include such an attribute line once resources are available (which is the case when Call Hold/Resume happens).
Note: it might be worthwhile to require only those attribute lines explicitly listed in C.21, step 5, and to not request any other lines that the UE sent previously.
3. The match against cr_SDP_Attribute_AnySendRecvMode will be successful for lines “a=sendrecv”, “a=recvonly”, “a=sendonly”, “a=inactive”. For the case of “a=sendrevc” (which is the case in the last SDP sent by the UE in PRACK or UPDATE), this needs to be replaced by “a=sendonly” (in case of Call Hold) or stay as is (in case of Call Resume).
4. In SDP last sent by UE, the UE did not know about the remote UE having resources. Hence, it specified “a=curr: qos remote none”. The remote UE resp SS then responded with “a=curr:qos local sendrecv”, signaling that it now has resources. Hence, the next SDP sent by UE (which is during Call Hold and then Call Resume) has to reflect this via “a=curr: qos remote sendrecv”.

5. According to approved CR R5-142951, the UE can send both “a=des:qos optional remote sendrecv” or “a=des:qos mandatory remote sendrecv” in PRACK or UPDATE during call setup. Especially, when the remote UE then responds with strength tag mandatory, it is likely that the UE will use such strength tag in follow-on SDP.

	Summary of change
	1. Added input parameter of type IMS_CallHoldResume_Type to argument list
2. When line “a=conf …” is encountered, do not include it in template for Call Hold or Resume. 
3. Made the distinction between Call Hold and Call Resume (and this is finally the reason why we need this information here). In case of Call Hold, it is ensured that “a=sendonly” is expected. For Call Resume, it is ensured that “a=sendrecv” is expected.
Note: this might fail if UE decides to signal Call Hold resp Call Resume on session level rather than on media level, see 24.610, section 4.5.2.1. To make this foolproof, we would need a more elaborate scheme to check that UE uses at least one of the two forms (session level or media level, “or” being inclusive).  
4. Replaced “a=curr:qos remote none” by “a=curr:qos remote sendrecv”. This is done via new helper function f_SDP_ExchangeAttribute2(), see Change 5
5. Allow both forms “a=des:qos optional remote sendrecv” and “a=des:qos mandatory remote sendrecv”, using new helper function f_SDP_ExchangeAttribute2()

	TTCN module
	 Common/IMS/IMS_SDP_Messages

	MCC160 Comment
	


Before change

    function fl_IMS_BuildSDP_AnnexC8_Step1_Attributes(template (omit) SDP_attribute_list p_AttributeList) return template SDP_attribute_list

  { /* remove direction attributes from the attribute list but change template of attribute list to superset

     => there can be any direction attribute in the first plcase => direction attribute to be checked later on  */

    var integer i;

    var integer k := 0;

    var SDP_attribute v_SDP_attribute;

    var SDP_attribute_list v_AttributeList;

    var template SDP_attribute_list v_NewAttributeList := omit;

    if (isvalue(p_AttributeList)) {

      v_AttributeList := valueof(p_AttributeList);

      for (i:=0; i < lengthof(v_AttributeList); i:=i+1) {

        v_SDP_attribute := v_AttributeList[i];

        if (not match(v_SDP_attribute, cr_SDP_Attribute_AnySendRecvMode)) {

          v_NewAttributeList[k] := v_SDP_attribute;

          k := k+1;

        }

      }

    }

    if (ispresent(v_NewAttributeList)) {

      v_NewAttributeList := superset(all from v_NewAttributeList);

    }

    return v_NewAttributeList;

  }
After change

  function fl_IMS_BuildSDP_AnnexC8_Step1_Attributes(template (omit) SDP_attribute_list p_AttributeList, IMS_CallHoldResume_Type p_HoldResume := HOLD) return template SDP_attribute_list  //Change 3.1
  { /* remove direction attributes from the attribute list but change template of attribute list to superset

     => there can be any direction attribute in the first plcase => direction attribute to be checked later on  */

    var integer i;

    var integer k := 0;

    var SDP_attribute v_SDP_attribute;

    var SDP_attribute_list v_AttributeList;

    var template SDP_attribute_list v_NewAttributeList := omit;

    if (isvalue(p_AttributeList)) {

      v_AttributeList := valueof(p_AttributeList);

      for (i:=0; i < lengthof(v_AttributeList); i:=i+1) {

        v_SDP_attribute := v_AttributeList[i];

        if (not match(v_SDP_attribute, cr_SDP_Attribute_AnySendRecvMode)) {

            if (not match(v_SDP_attribute, cr_SDP_Attribute_conf_qos)) { //Change 3.2
               v_NewAttributeList[k] := v_SDP_attribute;

               k := k+1;
            }

        }

        else {  //Change 3.3
          if ((p_HoldResume == HOLD)) {

            v_NewAttributeList[k] := cr_SDP_Attribute_sendonly;

          }

          else {  //RESUME

            v_NewAttributeList[k] := cr_SDP_Attribute_sendrecv;

          }
          k := k+1;

        }

      }

    }

    v_NewAttributeList := f_SDP_ExchangeAttribute2(v_NewAttributeList, cr_SDP_Attribute_curr_qos(c_remote, c_none), cr_SDP_Attribute_curr_qos(c_remote, c_sendrecv)); //Change 3.4
    v_NewAttributeList := f_SDP_ExchangeAttribute2(v_NewAttributeList, cr_SDP_Attribute_des_qos(c_optional, c_remote), cr_SDP_Attribute_des_qos((c_optional, c_mandatory), c_remote)); //Change 3.5
    if (ispresent(v_NewAttributeList)) {

      v_NewAttributeList := superset(all from v_NewAttributeList);

    }

    return v_NewAttributeList;

  }  
Change 4 – f_SDP_ExchangeAttribute2
	Function name
	 f_SDP_ExchangeAttribute2

	Reason for change
	Existing function  f_SDP_ExchangeAttribute could not be used because of the template restrictions in the argument list

	Summary of change
	 Provided a new function taking away the template restrictions

	TTCN module
	 Common/IMS/IMS_SDP_Templates

	MCC160 Comment
	


Before change

  n/a
After change

     function f_SDP_ExchangeAttribute2(template SDP_attribute_list p_AttributeList,

                                   template  SDP_attribute p_AttributeToBeChanged,

                                   template  SDP_attribute p_NewAttribute) return template SDP_attribute_list

  { // @sic R5s120530 change 3.17 sic@

    var template SDP_attribute_list v_AttributeList := p_AttributeList;

    var integer i := 0;

    if (isvalue(v_AttributeList)) {

      for (i:=0; i<lengthof(v_AttributeList); i:=i+1) {

        if (match(valueof(v_AttributeList[i]), p_AttributeToBeChanged)) {

          v_AttributeList[i] := p_NewAttribute;

        }

      }

    }

    return v_AttributeList;

  }
