
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 15.8 which is part of the ‘IMS_34229_IWD_14wk24’ test suite in ATS ‘iwd-TTCN3-B2013-03_D14wk24’.

The test case can be demonstrated to run with one LTE UE (see section 5). Execution log is provided as evidence.

3. Verification Test Summary

Test Case:
15.8
Test Group:
IMS Supplementary Services
ATS Version:
iwd-TTCN-B2013-03_D14wk24
System Simulator used:
Rohde & Schwarz CMW 500
UE used:
Nvidia Icera i500
Verification Status:
PASS
4. Corrections required
This section describes the TTCN changes required to ‘iwd-TTCN-B2013-03_D14wk24’ release.
Change 1 – fl_TC_15_8_Steps10_17
	Function name
	fl_TC_15_8_Steps10_17

	Reason for change
	3. When generating a new dialog – and just manipulating the allocated IMS_DialogInfo_Type element – we need to overwrite the RemoteTag field in there. Note that from the viewpoint of the SS, this is the local tag.
4. According to RFC 3261, CSeq is to be incremented for each new request within a dialog. When setting up a new dialog in this test case, the remote Cseq value needs to be reset to the one of the MO Invite such that the next request sent by the UE (which is PRACK in step 12) is checked to have incremented from the MO INVITE request. 

	Summary of change
	3. Replaced call f_IMS_PTC_ImsInfo_DialogSetRemoteTag() to by call to  f_IMS_PTC_ImsInfo_DialogSetLocalTag().

4. Added call to set back the corresponding CSeq value.

	TTCN module
	IMS/15/IMS_CC_SSTestcases

	MCC160 Comment
	


Before change

   function fl_TC_15_8_Steps10_17(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp) runs on IMS_PTC return template (present) SipUrl

  { /* return contact of the callee to which the call is forwarded to */

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_InviteRequestWithSdp.RoutingInfo);

    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;

    var SDP_Message v_SDP_MessageStep2 := p_InviteRequestWithSdp.SdpOffer;

    var charstring v_RemoteIPAddr := v_SDP_MessageStep2.origin.addr;  // the UE's IP address as received in the SDP offer

    var charstring v_ForwardedToCalleeContactUri := "sip:15.8@" & tsc_IMS_CallerDomain;

    var template (present) SipUrl v_NewContactUriRX := f_SIP_BuildSipUri_RX(v_ForwardedToCalleeContactUri);

    var template (value) SipUrl v_TargetedTo := f_SIP_BuildSipUri_TX("sip:user@company.com");

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (value) MessageHeader v_MessageHeader_Response;

    var template (present) MessageHeader v_MessageHeader_Request;

    var PRACK_Request v_PrackRequest;

    // 15.8 Step 10 Send 181 Call is being forwarded

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine181, f_IMS_InviteResponse_181_MessageHeaderTX(v_InviteRequest, v_TargetedTo))));

    // set new tag for the dialog:

    f_IMS_PTC_ImsInfo_DialogSetRemoteTag(f_IMS_GenerateTag("INVITE-FORWARDED"));

    // 15.8 Step 11 Send 183 Session in Progress

    v_MessageHeader_Response := f_IMS_InviteResponse_183_MessageHeaderTX(v_InviteRequest, tsc_OptionTagList_precondition, v_ForwardedToCalleeContactUri);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine183,

                                                                   v_MessageHeader_Response,

                                                                   cs_MessageBody_SDP(f_IMS_BuildSDP_AnnexC21_Step4(v_RemoteIPAddr, v_SDP_MessageStep2)))));

    // 15.8 Step 12 - 13B (same as steps 5 - 8 of C.21)

    f_IMS_MOCallSetup_AnnexC21_Steps5_8(p_InviteRequestWithSdp, v_MessageHeader_Response);    /* the procedure uses the (remote) contact address as in the 183 response (v_MessageHeader_Response)

                                                                                                 => in fact v_ForwardedToCalleeContactUri is used */

    // 15.8 Step 14. 180 Ringing

    v_MessageHeader_Response := f_IMS_InviteResponse_180_MessageHeaderTX(v_InviteRequest, v_ForwardedToCalleeContactUri);

    v_MessageHeader_Response.historyInfo := cs_HistoryInfo(v_TargetedTo);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine180, v_MessageHeader_Response)));

    // 15.8 Step 14A. Receive PRACK

    v_MessageHeader_Request := f_IMS_PrackRequest_MessageHeaderRX(v_InviteRequest, v_MessageHeader_Response, omit);

    IMS_Server.receive(car_IMS_Prack_Request(cr_PRACK_Request(v_NewContactUriRX, v_MessageHeader_Request))) -> value v_IMS_DATA_REQ;

    v_PrackRequest := v_IMS_DATA_REQ.Request.Prack;

    // 15.8 Step 14B. Send 200 OK for PRACK

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderTX(v_PrackRequest.msgHeader))));

    // Step 15. Send 200 OK for INVITE

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, -, -, v_ForwardedToCalleeContactUri);

    v_MessageHeader_Response.historyInfo := cs_HistoryInfo(v_TargetedTo);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_Response)));

    // Step 16. Finally, UE acknowledges call setup.

    v_MessageHeader_Request := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200);

    IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(v_NewContactUriRX, v_MessageHeader_Request)));

    return v_NewContactUriRX;

  }
After change

     function fl_TC_15_8_Steps10_17(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp) runs on IMS_PTC return template (present) SipUrl

  { /* return contact of the callee to which the call is forwarded to */

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(p_InviteRequestWithSdp.RoutingInfo);

    var INVITE_Request v_InviteRequest := p_InviteRequestWithSdp.Invite;

    var SDP_Message v_SDP_MessageStep2 := p_InviteRequestWithSdp.SdpOffer;

    var charstring v_RemoteIPAddr := v_SDP_MessageStep2.origin.addr;  // the UE's IP address as received in the SDP offer

    var charstring v_ForwardedToCalleeContactUri := "sip:15.8@" & tsc_IMS_CallerDomain;

    var template (present) SipUrl v_NewContactUriRX := f_SIP_BuildSipUri_RX(v_ForwardedToCalleeContactUri);

    var template (value) SipUrl v_TargetedTo := f_SIP_BuildSipUri_TX("sip:user@company.com");

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var template (value) MessageHeader v_MessageHeader_Response;

    var template (present) MessageHeader v_MessageHeader_Request;

    var PRACK_Request v_PrackRequest;

    // 15.8 Step 10 Send 181 Call is being forwarded

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine181, f_IMS_InviteResponse_181_MessageHeaderTX(v_InviteRequest, v_TargetedTo))));

    // set new tag for the dialog:

    f_IMS_PTC_ImsInfo_DialogSetLocalTag(f_IMS_GenerateTag("INVITE-FORWARDED")); 

    f_IMS_PTC_ImsInfo_CseqSet(dialogRemote, v_InviteRequest.msgHeader.cSeq.seqNumber);
    // 15.8 Step 11 Send 183 Session in Progress

    v_MessageHeader_Response := f_IMS_InviteResponse_183_MessageHeaderTX(v_InviteRequest, tsc_OptionTagList_precondition, v_ForwardedToCalleeContactUri);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine183,

                                                                   v_MessageHeader_Response,

                                                                   cs_MessageBody_SDP(f_IMS_BuildSDP_AnnexC21_Step4(v_RemoteIPAddr, v_SDP_MessageStep2)))));

    // 15.8 Step 12 - 13B (same as steps 5 - 8 of C.21)

    f_IMS_MOCallSetup_AnnexC21_Steps5_8(p_InviteRequestWithSdp, v_MessageHeader_Response);    /* the procedure uses the (remote) contact address as in the 183 response (v_MessageHeader_Response)

                                                                                                 => in fact v_ForwardedToCalleeContactUri is used */

    // 15.8 Step 14. 180 Ringing

    v_MessageHeader_Response := f_IMS_InviteResponse_180_MessageHeaderTX(v_InviteRequest, v_ForwardedToCalleeContactUri);

    v_MessageHeader_Response.historyInfo := cs_HistoryInfo(v_TargetedTo);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine180, v_MessageHeader_Response)));

    // 15.8 Step 14A. Receive PRACK

    v_MessageHeader_Request := f_IMS_PrackRequest_MessageHeaderRX(v_InviteRequest, v_MessageHeader_Response, omit);

    IMS_Server.receive(car_IMS_Prack_Request(cr_PRACK_Request(v_NewContactUriRX, v_MessageHeader_Request))) -> value v_IMS_DATA_REQ;

    v_PrackRequest := v_IMS_DATA_REQ.Request.Prack;

    // 15.8 Step 14B. Send 200 OK for PRACK

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, f_IMS_OtherResponse_200_MessageHeaderTX(v_PrackRequest.msgHeader))));

    // Step 15. Send 200 OK for INVITE

    v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, -, -, v_ForwardedToCalleeContactUri);

    v_MessageHeader_Response.historyInfo := cs_HistoryInfo(v_TargetedTo);

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_Response)));

    // Step 16. Finally, UE acknowledges call setup.

    v_MessageHeader_Request := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200);

    IMS_Server.receive(car_IMS_Ack_Request(cr_ACK_Request(v_NewContactUriRX, v_MessageHeader_Request)));

    return v_NewContactUriRX;

  }
5. Execution Log Files

Nvidia Icera i500 UE passed this test case on the Rohde & Schwarz CMW 500. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_15_8_Nvidia_Icera_i500.log

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.
6. References

[1] R5s14aaaa: Supporting information for re-verification of IMS test case 15.8 over LTE with 34.229-3 test model - This archive comprises html format execution log file
